
[bookmark: clause_Basic_Param_Value][bookmark: _Toc514234736]5.4.1	Formal parameters
[bookmark: _Toc514234737]5.4.1.0	General
TTCN-3 modules, structured types, templates, functions, altsteps, and testcases may be defined incompletely, i.e. some entities (variables, templates, ports, timers, etc.) used by the above objects may not be resolved in the definition of the object. These objects are called parameterized objects. Formal entities replacing the unresolved entities in the parameterized object's definition are called formal parameters.
Formal parameters of parameterized templates, functions, altsteps, and testcases are defined in formal parameter lists. Formal parameters of modules are defined in module parameter definitions (see clause 8.2.1).
Formal parameters shall be in, inout or out parameters (see definitions in clause 3.1). If not stated otherwise, a formal parameter is an in parameter. For all these three sorts of parameter passing, the formal parameters can both be read and set (i.e. get new values being assigned) within the parameterized object. Formal parameters can be used directly as actual parameters for other parameterized objects, e.g. as actual parameters in function invocations or as actual parameters in template instances.
If parameters are passed by value (i.e. in case of in and out parameters), type compatibility rules specified in clause 6.3 apply. When parameters are passed by reference, strong typing is required. Both the actual and formal parameter shall be of the same type.
Formal in parameters may have default values. This default value is used when no actual parameter is provided.
NOTE 1:	Although out parameters can be read within the parameterized object, they do not inherit the value of their actual parameter; i.e. they should be set before they are read.
Formal value or template parameters may be declared lazy using the @lazy modifier. The behaviour of lazy parameters is defined in clause 3.1, definition of lazy values or templates. See examples in clause 5.4.1.1.
Formal value or template parameters may be declared fuzzy using the @fuzzy modifier. The behaviour of lazy parameters is defined in clause 3.1, definition of fuzzy values or templates. See examples in clause 5.4.1.1.
Formal value or template parameters that are declared lazy or fuzzy can be additionally declared deterministic using the @deterministic modifier.
NOTE 2:	The actual values of component variables used in the delayed evaluation of a lazy or fuzzy parameter may differ from their values at the time, when the parameterized function or alstep was called.
Assigning default values for lazy and fuzzy formal parameters does not change the parameters' semantics: when the default values are used as actual values for the parameters, they shall be evaluated the same way (i.e. delayed) as if an actual parameter was provided.
Lazy and fuzzy properties are valid only in the scope, where the parameters' names are visible. For example, if a fuzzy parameter is passed to a formal parameter declared without a modifier, it loses its fuzzy feature inside the called function. Similarly, if it is passed to a lazy formal parameter, it becomes lazy within the called function.
[bookmark: _Toc514234738]5.4.1.1	Formal parameters of kind value
Values of all basic and user-defined types, address, component, port and timer types, and the default type can be passed as value parameters.
Syntactical Structure
[ ( in | inout | out ) ] [ ( @lazy | @fuzzy ) [@deterministic] ] 
Type ValueParIdentifier [ArrayDef] 
														[ ":=" ( Expression | "-" ) ]

Semantic Description
Value formal parameters can be used within the parameterized object the same way as values, for example in expressions.
Value formal parameters may be in, inout or out parameters. The default for value formal parameters is in parameterization which may optionally be denoted by the keyword in. Using of inout or out kind of parameterization shall be specified by the keywords inout or out respectively.
In parameters may have a default value, which is given by an expression assigned to the parameter. Formal parameters of modified templates may inherit the default values from the corresponding parameters of their parent templates; this shall explicitly be denoted by using a dash (don't change) symbol at the place of the modified template parameters' default value.
NOTE 1:	If functions are used for the initialization of default values of in parameters, it is strongly advised to avoid side effects during the evaluation of default values. Side effects may cause non-deterministic test executions. They can be avoided, e.g. by adhering to the rules defined in clause 16.1.4.
TTCN‑3 supports value parameterization according to the following rules:
the language element module allows static value parameterization to support test suite parameters, i.e. this parameterization may or may not be resolvable at compile-time but shall be resolved by the commencement of runtime (i.e. static at runtime). This means that, at runtime, module parameter values are globally visible but not changeable (see more details in clause 8.2);
the language elements template, testcase, altstep and function support dynamic value parameterization (i.e. this parameterization shall be resolved at runtime).
NOTE 2:	Component and default references are also handled as value parameters. In the case of component references, the corresponding component type is the type of the formal parameter. In the case of default references the TTCN-3 type default is the type of the formal parameter.
Restrictions
a)	Language elements which cannot be parameterized are: const, var, timer, control, record of, set of, enumerated, port, component and subtype definitions, group and import.
b)	Formal value parameters of templates, and of altsteps activated as defaults (see clause 20.5.2) shall always be in parameters.
c)	Restrictions on module parameters are given in clause 8.2.
d)	Default values can be provided for in parameters only.
e)	The expression of formal parameter's default value has to be compatible with the type of the parameter. The expression may be any expression that is well-defined at the beginning of the scope of the parameterized entity, but shall not refer to other parameters of the same parameter list.
f)	Default values of component type formal parameters shall be one of the special values null, mtc, self, or system.
g)	Default values of port, timer or default type formal parameters shall be the special value null.
h)	The dash (don't change) symbol shall be used with formal parameters of modified templates only (see also clause 15.5).
i)	For formal value parameters of templates the restrictions specified in clause 15 shall apply.
j)	Only in parameters can be declared lazy or fuzzy.
k)	When parameters are referenced (e.g. in assignments, expressions, template bodies, etc.), the rules for variables shall apply.
l)	Only function and altstep definitions with the exception of functions or altsteps started as test component behaviour (see clause 21.3.2) may have formal parameters of a port, timer or default type or of a type that contains a direct or indirect element or field of a port, default or timer type.
m)	Only function, altstep and testcase definitions may have formal parameters of a component type or of a type that contains a direct or indirect element or field of a component type.
n)	If a lazy or fuzzy value parameter is used in deterministic contexts (i.e. during the evaluation of a snapshot or initialization of global non-fuzzy templates), it shall be declared @deterministic.
Examples
EXAMPLE 1:	In, out and inout formal parameters
	function f_myFunction1(in boolean p_myReferenceParameter){ … };
	// p_myReferenceParameter is an in value parameter. The parameter can be read. It can also be
	// set within the function, however, the assignment is local to the function only
	
	function f_myFunction2(inout boolean p_myReferenceParameter){ … };
	// p_myReferenceParameter is an inout value parameter. The parameter can be read and set 
	// within the function - the assignment is not local
	
	function f_myFunction3(out template boolean p_myReferenceParameter){ … };
	// p_myReferenceParameter is an out value parameter. The parameter can be set within the 
	// function, the assignment is not local. It can also be read, but only after it has been set.

EXAMPLE 2:	Reading and setting parameters
	type record MyMessage {
		integer f1,
		integer f2
	}

	function f_myMessage (integer p_int) return MyMessage {
		var integer v_f1, v_f2;
		v_f1 := f_mult2 (p_int);
			// parameter p_int is passed on; as the parameter of the called function f_mult2 is
			// defined as an inout parameter, it passes back the changed value for p_int, 
		v_f2 := p_int;
		return {v_f1, v_f2};

	}

	function f_mult2 (inout integer p_integer) return integer {
		p_integer := 2 * p_integer;
			// the value of the formal parameter is changed; this new value is passed back when
			// f_mult2 completes
		return p_integer-1
	}

	testcase TC_01 () runs on MTC_PT {
	...
		p1.send (f_myMessage(5))
			// the value sent is { f1 := 9 , f2 := 10 }
	...
	}

EXAMPLE 3:	Function with default value for parameter
	function f_comp (in integer p_int1, in integer p_int2 := 3) return integer {
	  var integer v_v := p_int1 + p_int2;
	  return v_v;
	}

	function f_f () {
	  var integer v_w;
	  v_w := f_comp(1);   // same as calling f_comp(1,3);
	  v_w := f_comp(1,2); // value 2 is taken for parameter p_int2 and not its default value 3
	  …
	}

	
	type component Comp { var integer i := 0 }

	function g(integer x := f_comp(i)) runs on Comp return integer {
	    // reference to i from Comp is allowed in default value of parameter x
		return x;
	}

	function h(integer y := g()+i) runs on Comp {
		// reference to g is allowed because it has a compatible runs on clause as h
	}

EXAMPLE 4:	Direct passing of formal parameters to functions
	function f_myFunc2(in bitstring p_refPar1, inout integer p_refPar2) return integer {
  		:
	}
	function f_myFunc1(inout bitstring p_refPar1, out integer p_refPar2) return integer {
	  :
	  return f_myFunc2(p_refPar1, p_refPar2);
	}
	// p_refPar1 and p_refPar2 can be passed directly to a function invocation

EXAMPLE 5:	Lazy and fuzzy parameters
	type component MyComp { var integer vc_int }

	function f_MyLazyFuzzy(in @lazy integer p_lazy, in @fuzzy integer p_fuzzy) runs on MyComp {
	  //When called from MyCalling:
	  v_int := 1;
	  log(p_lazy); //will log 2 as function double with actual parameter vc_int equals 1 is called
	               //here; 2 is  stored in p_lazy (also, function double stores 2 in v_int)
	  log(p_lazy); //will log 2 again as p_lazy is not re-evaluated
	  log(p_fuzzy);//will log 4 as function double with actual parameter vc_int equals 2 is called
	               // here (also, function double stores 4 in vc_int)
	  log(p_fuzzy) //will log 8 as function double is re-evaluated with actual parameter 4
	}

	function f_double (in integer p_in) runs on MyComp return integer{
	  p_in := 2* p_in;
	  v_int := p_in;
	  return p_in
	}

	testcase TC_MyCalling() runs on MyComp {
	vc_int := 0;
	f_myLazyFuzzy (f_double(vc_int), f_double(vc_int) )
	}

EXAMPLE 6:	Difference between passing by value and passing by reference
	function f_byValue (in integer p_int1, in integer p_int2) {
	  p_int2 := p_int2 + 1;
	  log(p_int1);
	  log(p_int2);
	}

	function f_byReference (inout integer p_int1, inout integer p_int2) {
	  p_int2 := p_int2 + 1;
	  log(p_int1);
	  log(p_int2);
	}

	function f_f () {
	  var integer v_int := 1; 
	  f_byValue(v_int, v_int); // prints 1 and 2
	  log(v_int); // prints 1
	  f_byReference(v_int, v_int); // prints 2 and 2
	  log(v_int); // prints 2
	}

[bookmark: clause_Basic_Param_Template][bookmark: _Toc514234739]5.4.1.2	Formal parameters of kind template
Template kind parameters are used to pass templates into parameterizable objects.
Syntactical Structure
[ in | inout | out ] template [ Restriction ] [ ( @lazy | @fuzzy ) [@deterministic] ] 
Type ValueParIdentifier [ArrayDef] 
														 [ ":=" ( TemplateInstance | "-" ) ]

Semantic Description
Template parameters can be defined for templates, functions, altsteps, and test cases.
To enable a parameterized object to accept templates or matching symbols as actual parameters, the extra keyword template shall be added before the type field of the corresponding formal parameter. This makes the parameter a template parameter and in effect extends the allowed actual parameters for the associated type to include the appropriate set of matching attributes (see annex B) as well as the normal set of values.
Formal template parameters can be used within the parameterized object the same way as templates and template variables.
Formal template parameters may be in, inout or out parameters. The default for formal template parameters is in parameterization.
In parameters may have a default template, which is given by a template instance assigned to the parameter. Formal template parameters of modified templates may inherit their default templates from the corresponding parameters of their parent templates; this shall explicitly be denoted by using a dash (don't change) symbol at the place of the modified template parameter's default template. If a default template is used, it is evaluated in the scope of the parameterized entity, not the scope of the actual parameter list.
Formal template parameters can be restricted to accept actual parameters containing a restricted set of matching mechanisms only. Such limitations can be expressed by the restrictions omit, present, and value. The restriction template (omit) can be replaced by the shorthand notation omit. The meaning of the restrictions is explained in clause 15.8.
Restrictions
a)	Only function, testcase, altstep and template definitions may have formal template parameters.
b)	Formal template parameters of templates and of altsteps activated as defaults (see clause 20.5.2) shall always be in parameters.
c)	Default templates can be provided for in parameters only.
d)	The default template instance has to be compatible with the type of the parameter. The template instance may be any template expression that is well-defined at the beginning of the scope of the parameterized entity, but shall not refer to other parameters in the same parameter list. 
e)	Default templates of component type formal parameters shall be built from the special values null, mtc, self, or system.
f)	Restrictions specified in clause 15 shall apply.
g)	The dash (don't change) symbol shall be used with formal parameters of modified templates only (see also clause 15.5).
h)	Only in template parameters can be declared lazy or fuzzy.
i)	When template parameters are referenced (e.g. in assignments, expressions, template bodies, etc.), the rules for template variables shall apply.
j)	If a lazy or fuzzy template parameter is used in deterministic contexts (i.e. during the evaluation of a snapshot or initialization of global non-fuzzy templates), it shall be declared @deterministic.
[bookmark: _Toc514234740]5.4.2	Actual parameters
Values, templates, timers and/or ports can be passed into parameterized TTCN-3 objects as actual parameters. Actual parameters can be provided both as a list in the same order as the formal parameters as well as in an assignment notation explicitly using the associated formal parameter names or in a mixed notation where the first parameters are given in list notation and additional parameters in assignment notation.
Syntactical Structure
( Expression |												// for value parameter
  TemplateInstance |											// for template parameter
  "-" ) |														// to skip a parameter with default
  ParameterId ":=" ( Expression | TemplateInstance | TimerRef | Port ) )

Semantic Description
Actual parameters that are passed by value to in formal value parameters shall be variables, literal values, module parameters, constants, value variables, invocations of value returning (external) functions, formal value parameters (of in, inout or out parameterization) of the current scope or expressions composed of the above.
Actual parameters that are passed to out formal value parameters shall be (template) variables, formal (template) parameters (of in, inout or out parameterization) or references to elements of (template) variables or formal (template) parameters of structured types. Furthermore it is allowed to use the dash symbol "-" as an actual out parameter, signifying that a possible result for that parameter will not be passed back.
Actual parameters that are passed to inout  formal value parameters shall be variables or formal value parameters (of in, inout or out parameterization) or references to elements of variables or formal value parameters of structured types.
NOTE 1:	Reference to a string element cannot be passed by reference as string types are not structured types.
Actual parameters that are passed to in formal template parameters shall be literal values, module parameters, constants, variables, invocations of value or template returning (external) functions, formal value parameters (of in, inout or out parameterization) of the current scope or expressions composed of the above, as well as templates, template variables or formal template parameters (of in, inout or out parameterization) of the current scope.
Actual parameters that are passed to out formal template parameters shall be template variables, formal template parameters or references to elements of  template variables or formal template parameters of structured types. Furthermore it is allowed to use the dash symbol "-" as an actual out parameter, signifying that a possible result for that parameter will not be passed back.
Actual parameters that are passed to inout formal template parameters shall be template variables or formal template parameters (of in, inout or out parameterization) of the current scope or references to elements of template variables or formal template parameters of structured types.
When actual parameters that are passed to in formal value or template parameters contain a value or template reference, rules for using references on the right hand side of assignments apply. When actual parameters that are passed to inout and out formal value or template parameters contain a value or template reference, rules for using references on the left hand side of assignments apply.
The values of out formal parameters are passed to the actual parameters in the same order as is the order of formal parameters in the definition of the parameterized TTCN-3 object. The value is passed back to the actual parameter only if within the invoked object a value is assigned to it. If no value is assigned, the actual parameter remains unchanged when the invoked object completes.
It is allowed to pass elements of structured values or templates (record, set, record of, set of, union and anytype values or templates) by reference. Modification of parameters passed this way affects the original structured value or template. Before passing the actual parameter, the rules for referencing the element on the left hand side of assignments are applied, expanding the structured value so that the referenced element becomes accessible (see clauses 6.2 and 15.6 for more details).
NOTE 2:	Because inout parameters are passed by reference and component variables are effectively also accessed by reference within a called function or altstep, passing parts of a structured component variable as an actual inout parameter may have confusing effects inside the parameterized behaviour: changing either the inout parameter or the component variable may also change the other simultaneously, which might break the intended algorithm. For this reason, such situations should be avoided.
When a formal parameter is an out parameter or has been defined with a default value or template, respectively, then it is not necessary to provide an actual parameter. In such a case the default value or template is taken as actual parameter.
The actual parameters are evaluated in the order of their appearance. If for some formal parameters, no actual parameter has been provided, if they are out parameters, the dash symbol "-" and for in parameters their default values are taken. Default values are evaluated after the evaluation of the actual parameters and the order of their evaluation corresponds to their order in the formal parameter list.
NOTE 3:	If assignment notation has been used for the actual parameter list, the order of the evaluation of actual parameters may differ from the order of the parameters in the formal parameter list.
The empty brackets for instances of parameterized templates that have only parameters with default values are optional when no actual parameters are provided, i.e. all formal parameters use their default values.
Restrictions
a)	When using list notation, the order of elements in the actual parameter list shall be the same as their order in the corresponding formal parameter list. For each formal inout parameter and for each in parameter without a default there shall be an actual parameter. The actual parameter of a formal out parameter or in parameter with default value can be skipped by using dash "-" as actual parameter. An actual parameter can also be skipped by just leaving it out if no other actual parameter follows in the actual parameter list - either because the parameter is last or because all following formal parameters are out parameters or have default values and are left out. The number of actual parameters in the list notation shall not exceed the number of parameters in the formal parameter list.
b)	Void.
c)	When using assignment notation, each formal parameter shall be assigned an actual parameter at most once. For each assigned actual parameter there shall exist a corresponding formal parameter of the same name. For each formal parameter without default value, there shall be an actual parameter. In order to use the default value of a formal parameter, no assignment for this specific parameter shall be provided.
d)	For in formal parameters, the type of the actual parameter shall be compatible with the type of the formal parameter. For out formal parameters, the type of the formal parameter shall be compatible with the type of the actual parameter. Strong typing is required for inout formal (parameters passed by reference). For in formal template parameters, the template restriction of the actual parameter shall not be less restrictive than the one of the formal parameter. For out formal template parameters, the template restriction of the actual parameter shall not be more restrictive than the one of the formal parameter. For inout formal template parameters, the template restriction of the actual and the formal parameter shall be the same.
e)	Actual parameters passed to restricted formal template parameters shall obey the restrictions given in clause 15.8.
f)	All parameterized entities specified as an actual parameter shall have their own parameters resolved in the top‑level actual parameter list.
g)	If the formal parameter list of TTCN‑3 objects function, testcase, altstep or external function is empty, then the empty parentheses shall be included both in the declaration and in the invocation of that object. In all other cases the empty parentheses shall be omitted.
NOTE 4:	signature objects also have formal parameters, see clauses 15.2 and 22.3 for their handling.
h)	Void.
i)	Restrictions on parameters passed to altsteps are given in clauses 16.2.1 and 20.5.2.
j)	Unless specified differently in the relevant clause(s), actual parameters passed to in or inout formal parameters shall be at least partially initialized (for an exemption see e.g. clause 16.1.2 of the present document).
k)	Functions, called by actual parameters passed to fuzzy or lazy formal parameters of the calling function, shall not have inout or out formal parameters. The called functions may use other functions with inout or out parameters internally.
l)	Actual parameters passed to out and inout parameters shall not be references to lazy or fuzzy variables.
m)	Whenever a value or template of a record, set, union, record of, set of, array and anytype type is passed as an actual parameter to an inout parameter, none of the fields or elements of this structured value or template shall be passed as an actual parameter to another inout parameter of the same parameterized TTCN-3 object. This restriction applies recursively to all sub-elements of the structured value or template in any level of nesting.
n)	If two or more actual parameters passed to inout parameters of the same parameterized TTCN-3 object contain a reference to distinct alternatives of the same union or anytype value, an error shall be produced.
o)	If the mixed notation is used, no value list notation shall be used following the first assignment notation and the parameters given in assignment notation shall not assign parameters that already have an actual parameter given in list notation.
p)	Actual parameters passed to fuzzy or lazy formal parameters shall fulfill the restrictions imposed on content of functions used in special places given in 16.1.4.
Examples
EXAMPLE 1:	Formal and actual parameter lists have to match
	// A function definition with a formal parameter list 
	function f_myFunction(integer p_formalPar1, boolean p_formalPar2, bitstring p_formalPar3) { … }

	// A function call with an actual parameter list 
	f_myFunction(123, true,'1100'B);

	// A function call with assignment notation for actual parameters 
	f_myFunction(p_formalPar1 := 123, p_formalPar3 := '1100'B, p_formalPar2 := true);

EXAMPLE 2:	In parameters
	function f_myFunction(in template MyTemplateType p_myValueParameter){ … };
	// p_myValueParameter is in parameter, the in keyword is optional

	// A function call with an actual parameter
	f_myFunction(m_myGlobalTemplate);

EXAMPLE 3:	Inout and out parameters
	function f_myFunction(inout boolean p_myReferenceParameter){ … };
	// p_myReferenceParameter is an inout parameter
	
	// A function call with an actual parameter
	f_myFunction(v_myBooleanVariable);
	// The actual parameter can be read and set within the function 
	
	
	function f_myFunction(out template boolean p_myReferenceParameter){ … };
	// p_myReferenceParameter is an out parameter
	
	// A function call with an actual parameter
	f_myFunction(v_myBooleanVariable);
	// The actual parameter is initially unbound, but can be set and read within the function.
	f_myFunction(-); // the outcoming value is not assigned to a variable

	type record of integer RoI;
	
	function f_swapElements (inout integer p_int1, inout integer p_int2) {
	  var integer v_tmp := p_int1;
	  p_int1 := p_int2;
	  p_int2 := v_tmp;
	}
	
	function f_testReferences (inout RoI p_roi, inout integer p_elem) { … }
	:
	var RoI v_roi := { 0, 1, 2, 3, 4, 5 }; 
	f_swapElements(v_roi[0], v_roi[5]); // after the function call, v_roi is { 5, 1, 2, 3, 4, 0 }
	f_testReferences(v_roi, v_roi[2]); // produces an error as elements of v_roi are not allowed
		// to be passed by reference if the parent structure (v_roi) is passed by reference too.

	function f_changeAndIncrement(out integer p_e, in integer p_v, inout integer p_i) {
		p_i := p_i + 1;
		p_e := p_v;
	}
	:
	var integer v_i := 0;
	f_changeAndIncrement(v_roi[v_i], 3, v_i); // increments p_i, but still assigns 3 to v_roi[0]

EXAMPLE 4:	A side effect caused by passing part of a component variable as inout parameter
	type component MyComp {
		var ROI v_rec := { 0, 1 }
	}
	
	testcase TC() runs on MyComp {
		f_test(v_rec[1]) // passing 2nd element of component variable as inout parameter
		log(v_rec); //will log { 2 , 2 }
	}

	function f_test(inout integer p_int) runs on MyComp {
		v_rec := { 2 }; // now, isbound(p_int) == false
		p_int := 2; // now, v_rec == { 2, 2 }
	}

EXAMPLE 5:	Empty parameter lists
	// A function definition with an empty parameter list shall be written as 
	function f_myFunction(){ … }
	
	// and shall be called as
	f_myFunction();
	
	
	// A template definition with a default value for a formal parameter written as 
	template MyRecord m_mytemplate (integer p_myValue:= 1):= { … }
	
	// may be used without actual parameter list  (i.e. the default value is used)
	myPCO.send(m_mytemplate)

EXAMPLE 6:	Nested parameter lists
	// Given the message definition 
	type record MyMessageType 
	{	
		integer 	field1,
		charstring	field2,
		boolean		field3
	}

	// A message template might be 
	template MyMessageType mw_myTemplate(integer p_myValue) :=
	{	
		field1 := p_myValue,
		field2 := pattern "abc*xyz",
		field3 := true
	}

	// A test case parameterized with a template might be 
testcase TC_001(template MyMessageType p_rxMsg) runs on PTC1 system TS1 {
	:
	myPCO.receive(p_rxMsg); 
	}

	// When the test case is called in the control part and the parameterized template is 
	// passed as an actual parameter, the template's actual parameters shall be provided 
	control
	{	:
		execute(TC_001(mw_myTemplate(7)));
		:
	}

EXAMPLE 7:	A typical use case for lazy parameterization
	modulepar boolean PX_LOG_MESSAGE := true;

    function f_logMsg(@lazy charstring p_complex) {
      if (PX_LOG_MESSAGE) { 
        log(p_complex);
      }
    }

	function f_computeComplexMessage() return charstring {
		// some complicated computation
	}

    f_logMsg(f_computeComplexMessage()); // f_computeComplexMessage() is only invoked if
                                         // PX_LOG_MESSAGE is true

EXAMPLE 8:	Actual parameters passed to lazy and fuzzy formal parameters
	type record MyMessage { integer id, float number }

	type port MyPortType message { inout MyMessage }

	type component MyMTC {
	  var integer vc_id;
	  port MyPortType p;
	}

	testcase TC_shootingMessages () runs on MyMTC {
	  connect(self:p,self:p);
	  f_sendLazy({vc_id, rnd()}); //note that at this point vc_id is unintialized yet
	  f_sendFuzzy({vc_id, rnd()})
	}

	function f_sendLazy(@lazy MyMessage p_pdu) runs on MyMTC {
	  for (vc_id := 1; vc_id<9; vc_id:=vc_id+1){
	  p.send(p_pdu); // the actual parameter passed to the formal parameter p_pdu is evaluated only
	                 // in the first loop;let say rnd() returns 0.924946;
	                 // the message { 1, 0.924946 } is sent out 8 times
	  }
	  setverdict(pass,"messages has been sent out")
	}

	function f_sendFuzzy(@fuzzy MyMessage p_pdu) runs on MyMTC {
	  for (vc_id := 1; vc_id<9; vc_id:=vc_id+1){
	  p.send(pdu); // the actual parameter passed to the formal parameter p_pdu is evaluated in each 
	               // loop; let say rnd() returns 0.924946, 0.680497, 0.630836, 0.648681, 0.428501,
	               // 0.262539, 0.646990, 0.265262 in subsequent calls; the messages  1, 0.924946 },
	               // {{ 2, 0.680497 }, { 3, 0.630836 }, { 4, 0.648681 }, { 5, 0.428501 },
	               //  { 6, 0.262539 }, { 7, 0.646990 } and { 8, 0.265262 } are sent out in sequence
	  }
	  setverdict(pass,"messages has been sent out")
	}

EXAMPLE 9:	Order of out parameters
    function f_initValues (out integer p_par1, out integer p_par2) {
      p_par1 := 1; 
      p_par2 := 2;
    }

	function f_f(){
		var integer v_var1;
		f_initValues(p_par2 := v_var1, p_par1 := v_var1);
		// After this function call, v_var1 will contain 2, as parameters are assigned in
		// the same order as in the definition of the f_initValues function. Thus p_par1 is
		// assigned first to v_var1 and p_par2 after that overwriting the previous value.
	}

EXAMPLE 10:	Skipped actual parameters
    function f_skip (out integer p_par1, in integer p_par2 := 2) {
      p_par1 := 1 + p_par2; 
    }

	function f_f(){
		// the following statements all have the same semantics :
		f_skip (-,-); // p_par2 is initialized with default value 2 and 
                      // the result of p_par1 is not assigned to any variable
		f_skip (p_par1 := -, p_par2 := -);
		f_skip (p_par2 := -); // skip p_par1
		f_skip (-) ;  // skip p_par2 because it is the last
		f_skip () ;   // skip p_par1 because all following are also skipped
	}

EXAMPLE 11:	Mixed notation
	function f_mixed (out integer p_par1, in integer p_par2 := 2, inout integer p_par3) {
	  p_par1 := 1 + p_par2; 
	}

	function f_f(){
		var integer v := 0;
		// the following statements all have the same semantics:
		f_mixed(-,2,v);
		f_mixed(-,p_par2 := 2, p_par3 := v);
		f_mixed(-,-,p_par3 := v);
		f_mixed(-,p_par3 := v, p_par2 := 2);
		
		// not allowed:
		f_mixed(-,2,p_par3 := v, p_par2 := 5); // p_par2 is already assigned in list notation
	}

[bookmark: clause_Var][bookmark: _Toc514234852]11	Declaring variables
[bookmark: _Toc514234853]11.0	General
TTCN-3 variables are statically typed variables. Variables are either value variables to store values or template variables to store templates.
Variables can be of simple basic types, basic string types, structured types, special data types (including subtypes derived from these types) as well as address, component or default, port or timer types.
Variables can be declared and used in the module control part, test cases, functions and altsteps. Additionally, variables can be declared in component type definitions. These variables can be used in test cases, altsteps and functions which are running on a given component type. 
Variables can be declared lazy using the @lazy modifier.
Alternatively, variables can be declared fuzzy using the @fuzzy modifier. 
If a variable is declared fuzzy or lazy they can additionally declared with the @deterministic modifier to indicate that when used in a deterministic evaluation context, any evaluation of the variable would have no side effect and would yield the same result.
Lazy and fuzzy features are valid only in the scope, where the variables' names are visible. For example, if a fuzzy variable is passed to a formal parameter declared without a modifier, it loses its fuzzy feature inside the called function. Similarly, if it is passed to a lazy formal parameter, it becomes lazy within the called function.
Whenever a lazy or fuzzy variable is assigned, the TE is required to save the lexical environment (the set of directly or indirectly referenced values and templates) valid at the time of the assignment, so that it is possible to resolve the expression at the time of evaluation of the lazy or fuzzy value or template. If the assignment was made on a lower scope than the evaluation, saving the lexical environment extends lifetime of the referenced variables defined on that lower scope.
Example
	var @fuzzy integer v_fuzzy := 1;
	var integer v_var;
	var boolean v_condition := true;
	if (v_condition) {
		var integer v_local := 0;
		v_fuzzy := v_local;
		v_local := 10;
	}
	// although v_local is no longer valid at this point, v_fuzzy still evaluates to 10 because
	// the lexical environment is available to the fuzzy variable:
	v_var := v_fuzzy;

[bookmark: clause_Var_ValueVar][bookmark: _Toc514234854]11.1	Value variables
A TTCN-3 value variable stores values. It is declared by the var keyword followed by a type identifier and a variable identifier. An initial value can be assigned at variable declaration.
It may be used at the right hand side as well as at the left hand side of assignments, in expressions, following the return keyword in bodies of functions with a return clause in their headers and may be passed to both value and template-type formal parameters.
Syntactical Structure
var [ ( @lazy | @fuzzy ) [ @deterministic ] ] Type 
  { VarIdentifier [ ArrayDef ] [ ":=" Expression ] 
			{ [ "," ] VarIdentifier [ ArrayDef ] [ ":=" Expression ] }+ [ ";" ]

Semantic Description
A value variable associates a name with the location of a value. A value variable may change its value during test execution several times. A value can be assigned several times to a value variable. The value variable can be referenced multiple times in a TTCN-3 module.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:
a)	Expression shall be of type Type.
b)	Value variables shall store values only.
c)	Value variables shall not be declared or used in a module definitions part (i.e. global variables are not supported in TTCN‑3).
d)	Use of uninitialized value variables at other places than the left hand side of assignments, in return statements, or as actual parameters passed to formal parameters shall cause an error.
e)	The initialization or assignment of a fuzzy or lazy variable shall not contain function calls of functions with inout or out parameters. The called functions may use other functions with inout or out parameters internally.
[bookmark: _GoBack]f)	If a lazy or fuzzy value variable is are used in deterministic contexts (i.e. during the evaluation of a snapshot or initialization of global non-fuzzy templates), it shall be declared @deterministic and the expression assigned to the variable shall fulfill the restrictions imposed on content of functions used in special places given in 16.1.4.the same restrictions apply to all functions used in the value assigned to the variable as for functions described in clause 16.1.4.
g)	The expression assigned to a lazy or fuzzy variable might contain a direct or indirect reference to this variable. Evaluation of such an expression shall cause a dynamic error.
h)	Using the dot notation (see clauses 6.2.1.1, 6.2.2.1 and 6.2.5.1) and index notation (see clauses 6.2.7 and 6.2.7) for referencing a field, alternative or element of an address value, which actual value is null shall cause an error.
i)	The expression shall evaluate to a value, which is at least partially initialized.
Examples
	var integer v_myVar0;
	var integer v_myVar1 := 1;
	var boolean v_myVar2 := true, v_myVar3 := false;
	var @lazy integer v_myLazyVar1 := v_myVar1+1;
	var timer v_timer1;
	timer t_myTimer1;	v_myVar1 := 2;
	v_myVar1 := v_myLazyVar1; // v_myLazyVar1 evaluates to 2 + 1
	v_myLazyVar1 := v_myLazyVar1 + 1;
	v_myVar1 := v_myLazyVar1; // causes an error as v_myLazyVar1 references itself
	v_timer1 := t_myTimer1;

[bookmark: clause_Var_Templ][bookmark: _Toc514234855]11.2	Template variables
A TTCN-3 template variable stores templates. They are declared by the var template keyword followed by a type identifier and a variable identifier. An initial content can be assigned at declaration. In addition to values, template variables may also store matching mechanisms (see clause 15.7).
Template variables may be used on the right hand side as well as on the left hand side of assignments, following the return keyword in bodies of functions defining a template-type return value in their headers and may be passed as actual parameters to template-type formal parameters. It is also allowed to assign a template instance to a template variable or a template variable field.
Syntactical Structure
var template [ @lazy | @fuzzy ] [ restriction ] [ (@lazy | @fuzzy) [ @deterministic ] ]
  Type { VarIdentifier [ ArrayDef ] ":=" TemplateBody 
					 { [ "," ] VarIdentifier [ ArrayDef ] ":=" TemplateBody }+ [ ";" ]

Semantic Description
A template variable associates a name with the location of a template or a value (as every value is also a template). 
A template variable may change its template during test execution several times. A template or value can be assigned several times to a template variable. The template variable can be referenced multiple times in a TTCN-3 module.
The content of a template variable can be restricted to the matching mechanisms specific value and omit in the same way as formal template parameters, see clause 5.4.1.2. The restriction template (omit) can be replaced by the shorthand notation omit.
NOTE 1:	String and list type templates can be concatenated, see clause 15.11.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:
a)	Template variables shall not be declared or used in a module definitions part (i.e. global variables are not supported in TTCN‑3).
[bookmark: OLE_LINK5][bookmark: OLE_LINK6]b)	When used on the right hand side of assignments template variables shall not be operands of TTCN‑3 operators (see clause 7.1) and the variable on the left hand side shall be a template variable too.
c)	When accessing element of template variables either on the left hand side or on the right hand side of assignments, the rules given in clause 15.6 shall apply.
NOTE 2:	While it is not allowed to directly apply TTCN‑3 operations to template variables, it is allowed to use the dot notation and the index notation to inspect and modify template variable fields.
d)	Use of uninitialized template variables at other places than the left hand side of assignments, in return statements, or as actual parameters passed to formal parameters shall cause an error.
e)	Void.
f)	If the template variable is restricted, then the template used to initialize it shall contain only the matching mechanisms as described in clause 15.8.
g)	Template variables, similarly to global and local templates, shall be fully specified in order to be used in sending and receiving operations.
h)	Restrictions on templates in clause 15 shall apply.
i)	The initialization or assignment of a fuzzy or lazy variable shall not contain function calls of functions with inout or out parameters. The called functions may use other functions with inout or out parameters internally.
j)	If a lazy or fuzzy template variable is are used in deterministic contexts (i.e. during the evaluation of a snapshot or initialization of global non-fuzzy templates), it shall be declared @deterministic and the same restrictions apply to all functions used in the template body assigned to the variable shall fulfill the restrictions imposed on content of functions used in special places given in 16.1.4.as for functions described in clause 16.1.4.
k)	Using the dot notation (see clauses 6.2.1.1, 6.2.2.1 and 6.2.5.1) and index notation (see clause 6.2.7) for referencing a field, alternative or element of an address value, which actual value is null shall cause an error.
l)	The template body at the right-hand side of the  assignment symbol shall evaluate to a value or template, which is type compatible with the variable being declared.
m)	The template body at the right-hand side of the  assignment symbol shall evaluate to an object that is at least partially initialized.
[bookmark: clause_Templates][bookmark: _Toc514234859]15	Declaring templates
[bookmark: _Toc514234860]15.0	General
Templates are used to either transmit a set of distinct values or to test whether a set of received values matches the template specification. Templates can be defined globally or locally.
Templates provide the following possibilities:
1. they are a way to organize and to re-use test data, including a simple form of inheritance;
1. they can be parameterized;
1. they allow matching mechanisms;
1. they can be used with either message-based or procedure-based communications.
Within a template values, ranges and matching attributes can be specified and then used in both message-based and procedure-based communications. Templates may be specified for any TTCN‑3 type or procedure signature. The type‑based templates are used for message-based communications and the signature templates are used in procedure‑based communications.
A template can be declared fuzzy using the @fuzzy modifier.
NOTE 1:	Using a fuzzy template from a non-fuzzy template causes evaluation of the fuzzy template. Thus, for unparameterized non-fuzzy templates, the result of the used fuzzy templates will stay the same for every usage.
A fuzzy template can be declared deterministic using the @deterministic modifier. A deterministic template shall be evaluated to the same result in the same deterministic evaluation context whenever it is evaluated. 
A modified template declaration (see clause 15.5) specifies only the fields to be changed from the base template, i.e. it is a partial specification.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:
1. Templates shall not be of default, port or timer type.
1. Templates shall not be of a structured type that contains fields of default, port or timer type on any level of nesting.
NOTE 2:	The anytype type does not include the default type, port and timer types (see clause 6.2.6), so that restriction b) does not apply to anytype templates.
1. The expression or template body initializing a template shall evaluate to a value or template, which is type compatible with the template being declared.
1. The expression or template body initializing a template shall evaluate to a value or a template that is at least partially initialized or to a matching mechanism.
1. The body of a fuzzy template shall not contain function calls of functions with inout or out parameters. The called functions may use other functions with inout or out parameters internally.
1. Fuzzy features are valid only in the scope, where the templates' names are visible. For example, if a fuzzy template is passed to a formal template parameter declared without a modifier, it loses its fuzzy feature inside the called function.
1. For a fuzzy template that is declared deterministic the template body shall fulfill the restrictions imposed on content of functions used in special places given in 16.1.4. 
[bookmark: clause_Templates_GlobalLocal][bookmark: _Toc514234863]15.3	Global and local templates
TTCN-3 allows defining global templates and local templates.
Syntactical Structure
template [ restriction ] [ @fuzzy [ @deterministic ] ] Type TemplateIdentifier  
["(" TemplateFormalParList ")"] 
[ modifies TemplateRef ] ":=" TemplateBody

NOTE:	The optional restriction part is covered by clause 15.8.
[bookmark: clause_Templates_Modifiedl][bookmark: _Toc514234865]15.5	Modified templates
In cases where small changes are needed to specify a new template, it is possible to specify a modified template. A modified template specifies modifications to particular fields of the original template, either directly or indirectly. As well as creating explicitly named modified templates, TTCN‑3 allows the definition of in-line modified templates.
Syntactical Structure
Global or local modified template:
template [restriction] [ @fuzzy [ @deterministic ]] Type TemplateIdentifier  
["(" TemplateFormalParList ")"] 
modifies TemplateRef ":=" TemplateBody

[bookmark: _16.1.4_Invoking_functions][bookmark: clause_FuncAltTC_Func_SpecificPlaces][bookmark: _Toc514234890]16.1.4	Invoking functions from specific places
If value returning functions are called in receiving communication operations (in templates, template fields, in-line templates as actual parameters or when evaluating the port expression), in timeout operations (when evaluating the timer expression), in test component operations (in guards or events of alt statements or altsteps, see clause 20.2), or in initializations of altstep local definitions (see clause 16.2), the following operations shall not be used present in functions called in the cases specified above, in order to avoid side effects that cause changing the state of the component or the actual snapshot and to prevent different results of subsequent evaluations on an unchanged snapshot:
1. All component operations, i.e. create, start (component), stop (component), kill, running (component), alive, done and killed (see notes 1, 3, 4 and 6).
1. All port operations, i.e. start (port), stop (port), halt, clear, checkstate, send, receive, trigger, call, getcall, reply, getreply, raise, catch, check, connect, disconnect, map and unmap (see notes 1, 2, 3, 4 and 6).
1. The action operation (see notes 2 and 6).
1. All timer operations, i.e. start (timer), stop (timer), running (timer), read, timeout (see notes 4 and 6).
1. Calling non-deterministic external functions, i.e. external functions where the resulting values for actual inout or out parameters or the return value may differ for different invocations with the same actual in and inout parameters (see notes 4 and 6).
1. Calling the rnd predefined function (see notes 4 and 6).
1. Changing of component variables, i.e. using component variables on the left-hand side of assignments, and in the instantiation of out and inout parameters (see notes 4 and 6).
1. Calling the setverdict operation (see notes 4 and 6).
1. Activation and deactivation of defaults, i.e. the activate and deactivate statements (see notes 5 and 6).
1. Calling functions and deterministic external functions with out or inout parameters (see notes 7 and 8).
1. Calling functions and external functions with @fuzzy formal parameters and variables (see notes 4 and 9).
1. The setencode operation (see note 8 and clause 27.9).
1. Referencing lazy or fuzzy variables, parameters or templates that have not been declared deterministic.
[bookmark: clause_Basic_Return][bookmark: _Toc514234910]19.10	The Return statement
The return statement terminates execution of functions or altsteps.
Syntactical Structure
return [ Expression | TemplateInstance ]

Semantic Description
The return statement terminates execution of a function or altstep and returns control to the point from which the function or altstep was called. When used in functions, a return statement may be optionally associated with a return value or template.
TTCN-3 allows optional statement blocks that may follow altstep calls within alt statements. If there is a statement block, the return statement returns control to the beginning of this statement block and the statement block is executed before the alt statement is left. If there is no statement block, test execution continues with the first statement following the alt statement.
The return value or template is first evaluated before returning.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5 and shown in table 16, the following restrictions apply:
1. The return statement shall not be used present in testcase definitions.
[bookmark: clause_Statements_Interleave][bookmark: _Toc514234920]20.4	The Interleave statement
The interleave statement allows to specify the interleaved occurrence and handling of receiving events including done, killed, timeout, receive, trigger, getcall, getreply, catch and check.
Syntactical Structure
interleave "{" 
		{ "[]" ( TimeoutStatement |
              ReceiveStatement |
              TriggerStatement |
              GetCallStatement |
              CatchStatement |
              CheckStatement |
              GetReplyStatement |
              DoneStatement |
              KilledStatement ) StatementBlock
		} 
"}"

Semantic Description
The interleave statement allows to specify the interleaved occurrence and handling of the statements done, killed, timeout, receive, trigger, getcall, getreply, catch and check.
Interleaved behaviour can always be replaced by an equivalent set of nested alt statements. The procedures for this replacement and the operational semantics of interleaving are described in part 4 of the TTCN‑3 standard (ETSI ES 201 873‑4 [1]).
The rules for the evaluation of an interleaving statement are the following:
1. Whenever a reception statement is executed, the following non-reception statements are subsequently executed until the next reception statement is reached, a break statement is reached, or the interleaved sequence ends.
NOTE 1:	Reception statements are TTCN‑3 statements which may occur in sets of alternatives, i.e. receive, check, trigger, getcall, getreply, catch, done, killed and timeout. Non-reception statements denote all other non-control-transfer statements which can be used within the interleave statement.
If none of the alternatives of the interleave statement can be executed, the default mechanism will be invoked. This means, according to the semantics of the default mechanism, the actual snapshot will be used to evaluate those altsteps that have been activated before entering the interleave statement.
NOTE 2:	The complete semantics of the default mechanism within an interleave statement is given by replacing the interleave statement by an equivalent set of nested alt statements. The default mechanism applies for each of these alt statements.
The evaluation then continues by taking the next snapshot if no break statement was encountered.
The evaluation of the interleave statement is terminated if a break statement is executed.
The operational semantics of interleaving are fully defined in part 4 of the TTCN‑3 standard (ETSI ES 201 873‑4 [1]).
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5 and shown in table 16, the following restrictions apply:
1. Control transfer statements activate, deactivate,  repeat, all calls of altsteps and (direct and indirect) calls of user-defined functions, which include reception statements, shall not be used present in interleave statements.
1. In addition, it is not allowed to guard branches of an interleave statement with Boolean expressions (i.e. the '[ ]' shall always be empty). It is also not allowed to specify else branches in interleaved behaviour.
1. An interleave statement used within the module control part shall only contain timeout statements.
1. The restricted use of the control transfer statements for, while, do-while, and goto within interleave statements is allowed under the following conditions:
28. The loop statements for, while, and do-while can be used within statements blocks that do not contain reception statements.
28. The goto statement can be used for defining jumps with statements blocks that do not contain reception statements and for specifying jumps out of interleave statements.
[bookmark: clause_ConfigOps][bookmark: _Toc514234926]21	Configuration Operations
[bookmark: _Toc514234927]21.0	General
Configuration operations are used to set up and control test components and their connections. They are summarized in table 20.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5, the following restrictions apply:
1. These operations shall only be used in:
TTCN‑3 test cases;
behaviours invoked directly or indirectly from a test case or from a behaviour started on a ptc.
They shall not be used present in:
the module control part;
functions or altsteps invoked directly or indirectly from the module control part;
declarations inside component type definitions; or
functions invoked directly or indirectly from declarations inside component type definitions.
[bookmark: tab_Config_Oper]Table 20: Overview of TTCN‑3 configuration operations
	Operation
	Explanation
	Syntax Examples

	Connection Operations

	connect
	Connects the port of one test component to the port of another test component
	connect(ptc1:p1, ptc2:p2);

	disconnect
	Disconnects two or more connected ports
	disconnect(ptc1:p1, ptc2:p2);

	map
	Maps the port of one test component to the port of the test system interface
	map(ptc1:q, system:sutPort1);

	unmap
	Unmaps two or more mapped ports
	unmap(ptc1:q, system:sutPort1);

	Test Component Operations

	create
	Creation of a normal or alive test component, the distinction between normal and alive test components is made during creation
(MTC behaves as a normal test component)
	Non-alive test components:
var PTCType c := PTCType.create;
Alive test components:
var PTCType c := PTCType.create alive;

	start
	Starting test behaviour on a test component, starting a behaviour does not affect the status of component variables, timers or ports
	c.start(PTCBehaviour());

	stop
	Stopping test behaviour on a test component
	c.stop;

	kill
	Causes a test component to cease to exist
	c.kill;

	alive
	Returns true if the test component has been created and is ready to execute or is executing already a behaviour; otherwise returns false
	if (c.alive) …

	running
	Returns true as long as the test component is executing a behaviour; otherwise returns false
	if (c.running) …

	done
	Checks whether the function running on a test component has terminated
	c.done;

	killed
	Checks whether a test component has ceased to exist
	c.killed { … }

	Test Case Operations

	stop
	Terminates the test case with the test verdict error
	testcase.stop ( … );

	Reference Operations

	mtc
	Gets the reference to the MTC
	connect(mtc:p, ptc:p);

	system
	Gets the reference to the test system interface
	map(c:p, system:sutPort);

	self
	Gets the reference to the test component that executes this operation
	self.stop;



[bookmark: _Toc514235032]A.1.6.1.3	Template definitions
[bookmark: TTemplateDef]TemplateDef ::= TemplateKeyword [TemplateRestriction] [FuzzyModifier [DeterministicModifier]]   
                    BaseTemplate [DerivedDef] AssignmentChar TemplateBody 
[bookmark: TBaseTemplate]BaseTemplate ::= (Type | Signature) Identifier ["(" TemplateOrValueFormalParList   
                                                    ")"] 

[bookmark: _Toc514235044]A.1.6.3	Local definitions
[bookmark: _Toc514235045]A.1.6.3.1	Variable instantiation
[bookmark: TVarInstance]VarInstance ::= VarKeyword (([(LazyModifier | FuzzyModifier) [DeterministicModifier]] 
                                 Type VarList) | 
                                 ((TemplateKeyword | RestrictedTemplate)   
                                 [(LazyModifier | FuzzyModifier) [DeterministicModifier]] 
                                 Type TempVarList)) 
[bookmark: TVarList]VarList ::= SingleVarInstance {"," SingleVarInstance} 
[bookmark: TSingleVarInstance]SingleVarInstance ::= Identifier [ArrayDef] [AssignmentChar Expression] 
[bookmark: TVarKeyword]VarKeyword ::= "var" 
[bookmark: TTempVarList]TempVarList ::= SingleTempVarInstance {"," SingleTempVarInstance} 
[bookmark: TSingleTempVarInstance]SingleTempVarInstance ::= Identifier [ArrayDef] [AssignmentChar TemplateBody] 
[bookmark: TValueRef][bookmark: TVariableRef]ValueRef ::= Identifier [ExtendedFieldReference] 

[bookmark: _Toc514235054]A.1.6.7	Parameterization
[bookmark: TInParKeyword]InParKeyword ::= "in" 
[bookmark: TOutParKeyword]OutParKeyword ::= "out" 
[bookmark: TInOutParKeyword]InOutParKeyword ::= "inout" 
[bookmark: TFormalValuePar]FormalValuePar ::= [(InParKeyword | 
                          InOutParKeyword | 
                          OutParKeyword 
                         )] 
                        [(LazyModifier | FuzzyModifier) [DeterministicModifier]] 
                        Type Identifier [ArrayDef] 
                        [":=" (Expression | Minus)] 
[bookmark: TFormalTemplatePar]FormalTemplatePar ::= [(InParKeyword | 
                             OutParKeyword | 
                             InOutParKeyword 
                            )] 
                          (TemplateKeyword | RestrictedTemplate) 
                               [(LazyModifier | FuzzyModifier) [DeterministicModifier]] 
                           Type Identifier [ArrayDef] [":=" (TemplateInstance | Minus)] 
[bookmark: TRestrictedTemplate]RestrictedTemplate ::= OmitKeyword | (TemplateKeyword TemplateRestriction) 
[bookmark: TTemplateRestriction]TemplateRestriction ::= "(" (OmitKeyword | 
                                  ValueKeyword | 
                                  PresentKeyword 
                                 ) ")" 

